

```

        TITL 'c99 CLI call function'
        IDT 'callcli'
*
* for use with c99/MDOS
*
* (c) 1990 Clint Pulley
*
        DEF CALLCL
*
* status=callcli(command_string);
*
* calls the MDOS Command Line Interpreter (CLI) to process the command
* in command_string (char[]). At MDOS 0.97h, only intrinsic commands may
* be processed (eg. dir, del, time etc.). Returns true if the running
* version of MDOS does not support this feature.
*
* CAUTION : Use of certain CLI commands (ie copy) with MDOS 1.14/0.97h
* will result in the system crashing when batch files are executed.
* This is believed to be an internal MDOS problem.
*
* Last edit 2320 90/05/09
*
CALLCL CLR  R8          FOR STATUS RETURN
        MOV  R14,R1     GET CURRENT STACK PTR
        AI   R1,-104    USE STACK FOR PAD SAVE AREA
        MOV  R1,@PADBF1 SAVE PTR IN CALLS
        MOV  R1,@PADBF2
        BL  @MOVEM      SAVE >F018->F07F
        DATA >F018
PADBF1 DATA 0
        DATA 104
        MOV  @2(R14),@CMDPTR
        BL  @MOVEM      MOVE COMMAND STRING TO PAD
CMDPTR DATA 0
        DATA >F020,81
        MOVB @>F117,@USRP7 SWAP PAGE 7
        MOVB @CLIP7,e>F117
        LI   R0,>F020   COMMAND
        LI   R1,>F208   CLI COMMAND BUFFER
        MOVB @NULL,*R1+ STORE NULL BYTE COUNT
LOOP1  MOVB *R0+,*R1+  COPY COMMAND
        JNE  LOOP1
        DEC  R1
        MOVB @SPACE,*R1+ TERMINATE WITH SPACE
        MOVB @NULL,*R1  AND NULL
        LI   R0,>F209   PTR TO START OF COMMAND
        BL  @MOVEM      SAVE MAPPER REGISTERS
        DATA >F110,>F200,7
        BL  @MOVEM      MOVE CLI CALL CODE TO PAGE 7
        DATA CCODE,>F300,CCSIZE
        BLWP @CALLC     CALL COMMAND PROCESSOR
        INC  R8         IF WRONG MDOS VERSION
        MOVB @USRP7,@>F117 RESTORE PAGE 7
        BL  @MOVEM      RESTORE >F018->F07F

```

```

PADBF2 DATA 0,>F018,104
      B      *R13
*
*      BL      @MOVEM      MOVE N BYTES
*      DATA FROM,TO,N
*
* USES R4, R5, R6
*
MOVEM MOV *R11+,R4      FROM
      MOV *R11+,R5      TO
      MOV *R11+,R6      N
MOVEM1 MOVB *R4+,*R5+
      DEC R6
      JNE MOVEM1
      RT
*
* CLI CALL CODE - RELOCATED TO >F300
*
CCODE LIM1 0
      MOV *R13,R1      START OF COMMAND
      MOV @2(R13),R2    END OF COMMAND
      CLR R0
      MOV R0,@>F110    MAP SYSTEM PAGE 0
      MOVB @>1B03,@>F110 MAP CLI PAGE 0
      MOV @>0402,R0
      CI R0,>ECEC      CLI CHECKWORD
      JNE CCERR      IF WRONG MDOS VERSION
      INCT R14      BYPASS ERROR EXIT
      MOV @>0404,R3    ADDRESS OF COMMAND PRCESSOR
      LI R4,>0114
      LI R5,>F110
      MOV *R4+,*R5+    MAP IN CLI
      MOV *R4+,*R5+
      MOV *R4+,*R6+
      MOVB *R4,*R5
      LI R0,>FE00
      MOVB R0,@>0102    TWEAK SCHFDILER
      BLWP *R3          CALL COMMAND PROCESSOR
CCERR LI R4,>F200
      LI R5,>F110
      MOV *R4+,*R5+    MAP IN CALLER
      MOV *R4+,*R5+
      MOV *R4+,*R5+
      MOVB *R4,*R5
      RTWP
CCSIZE EQU $-CCODE
*
CALLC DATA >F060,>F300 BLWP VECTOR FOR CLI CALL
USRP7 BYTE 0            USER PAGE 7
CLIP7 BYTE >13         CLI PAGE 7
NULL  BYTE 0
SPACE BYTE >20
*
      END

```